﻿/**
 * The AudioTrack Library by Chris J. Rock. Aug 10, 2011
 * Check out chrisjrock.net and be a king among men.
 *
 * Copyright (c) 2011 Chris J. Rock
 * 
 * Permission is hereby granted, free of charge, to any person
 * obtaining a copy of this software and associated documentation
 * files (the "Software"), to deal in the Software without
 * restriction, including without limitation the rights to use,
 * copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the
 * Software is furnished to do so, subject to the following
 * conditions:
 * 
 * The above copyright notice and this permission notice shall be
 * included in all copies or substantial portions of the Software.
 * 
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,
 * EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES
 * OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND
 * NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT
 * HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY,
 * WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING
 * FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR
 * OTHER DEALINGS IN THE SOFTWARE.
 **/


package net.chrisjrock.sound.audioTrack.mixer 
{
	import flash.utils.Dictionary;
	
	import net.chrisjrock.sound.audioTrack.track.IAudioTrack;
	
	/**
	 * A mixer allows multiple tracks to be treated as a single track for easy management.
	 * @author Christopher J. Rock
	 */
	public class AudioMixer extends AbstractMixer
	{
		protected var _tracks:Dictionary = new Dictionary();
		
		public function AudioMixer(trackId:String) 
		{
			super(trackId);
		}
		
		/**
		 * @inheritDoc
		 */		
		override public function dispose():void 
		{
			if (_tracks)
			{
				for (var id:String in _tracks)
				{
					removeTrack(_tracks[id]);
				}
				_tracks = null;
			}
			
			super.dispose();
		}
		
		/**
		 * @inheritDoc
		 */		
		override public function disposeAll():void 
		{
			if (_tracks)
			{
				var track:IAudioTrack;
				for (var id:String in _tracks)
				{
					track = _tracks[id];
					if (_tracks[id] is IAudioMixer)
					{
						IAudioMixer(track).disposeAll();
					}
					else
					{
						track.dispose();
					}
				}
				_tracks = null;
			}
			
			super.dispose();
		}
		
		/**
		 * @inheritDoc
		 */		
		override public function hasTrack (track:IAudioTrack):Boolean
		{
			if (_tracks == null)
			{
				throw (new Error("Track list has been destroyed."));
			}
			
			return _tracks[track.trackId] != null;
		}
		
		override protected function addTrackToCollection (track:IAudioTrack):void
		{
			if (_tracks == null)
			{
				throw (new Error("Track list has been destroyed."));
			}
			
			_tracks[track.trackId] = track;
		}
		
		override protected function removeTrackFromCollection (track:IAudioTrack):void
		{
			if (_tracks == null)
			{
				throw (new Error("Track list has been destroyed."));
			}
			
			delete _tracks[track.trackId];
		}
		
	}
	
}